Un po' di Analisi con Maxima

Andrea Centomo

Liceo "F. Corradini" di Thiene (VI)

Email: `andrea.centomo@istruzione.it`

25 agosto 2006

In queste pagine vengono descritti un certo numero di problemi classici dell'Analisi. Lo studio avviene con l'ausilio del computer algebra system libero Maxima.

1. Numeri Reali

Sia a un numero reale qualsiasi. Una rappresentazione decimale del numero a può essere di tre tipi:

1. finita: quando il numero di cifre decimali è un numero naturale;
2. infinita periodica: quando le cifre decimali sono infinite ma, da un certo punto in poi, si ripetono;
3. infinita aperiodica: quando non è verificato nessuno dei due casi precedenti.

Esempi di numeri dei primi due tipi sono rispettivamente:

$$\frac{487}{20} = 24.35 \qquad\qquad -\frac{25}{7} = -3.571428571428... = -3.\overline{571428}$$

mentre un numero del terzo tipo è il numero irrazionale $\sqrt{21} = 4.58257...$.

Un elaboratore, per quanto potente esso sia, è in grado di rappresentare **solamente** numeri decimali finiti. Per questa ragione, quando in esso vengono rappresentati dei numeri periodici o irrazionali, ciò avviene **sempre** in modo **approssimato**. Vediamo ora come Maxima tratti le approssimazioni di numeri periodici e irrazionali.

(C1) `-25/7;`

(D1) $-\frac{25}{7}$

(C2) `%,numer;`

(D2) -3.571428571428572

(C3) `sqrt(21);`

(D3) $\sqrt{21}$

(C4) `%,numer;`

(D4) 4.58257569495584

Da un punto di vista sintattico osserviamo innanzitutto come la *radice quadrata* venga definita in Maxima attraverso il comando `sqrt` (square root). Il programma usualmente mantiene la rappresentazione simbolica dei numeri reali. Tuttavia, attraverso il comando `numer`, è possibile ottenere la rappresentazione decimale di un qualsiasi numero reale. La sintassi del comando `numer` è particolare: prima si indica il riferimento ad un output (nel nostro caso `%` si riferisce all'ultimo output restituito dal programma prima di eseguire il comando ossia rispettivamente a `D1` e `D3`), quindi si interpone una virgola e finalmente si digita il comando `numer`.

Risulta interessante osservare che nel primo caso vengono rappresentate 15 cifre decimali e che la quindicesima non è significativa, mentre nel secondo caso vengono rappresentate solo le prime 14 cifre significative.

Esercizio 1. Calcolare con Maxima la rappresentazione decimale dei numeri $223/21$ e $\sqrt{277}$.

Esercizio 2. Calcolare la rappresentazione decimale .

In molti casi è di interesse disporre di una rappresentazione decimale con precisione arbitraria ma fissata. Vediamo come esempio la seguente approssimazione del numero irrazionale π, che in Maxima si rappresenta con `%pi`:

```
(C5) %pi;
```

(D5) π

```
(C6) %,numer;
```

(D6) 3.141592653589793

```
(C7) fpprec:5;
```

(D7) 5

```
(C8) bfloat(D5);
```

```
3.1416B0
```

Il comando `fpprec` (floating point precision) fissa come valore della precisione 5 mentre il comando `bfloat` permette di valutare esplicitamente l'approssimazione. Osserviamo che Maxima esegue un'approssimazione alla quinta cifra decimale. Come ulteriore esempio calcoliamo con la medesima precisione un'approssimazione del numero:

$$\sqrt{7}+\sqrt{21}-\sqrt{11}$$

```
(C9) bfloat(sqrt(7)+sqrt(21)-sqrt(11));
```

```
3.9117B0
```

Esercizio 3. Calcolare un'approssimazione, con precisione alla settima cifra decimale, del numero irrazionale $\sqrt{131}+\sqrt{24}-\sqrt{13}$.

Esercizio 4. Calcolare un'approssimazione, con approssimazione alla nona cifra decimale, del numero di Nepero e che in Maxima si indica con `%e`.

2. Limiti

Attraverso il comando `limit` è possibile calcolare limiti sia nel caso di punti limite finiti che infiniti. Consideriamo come esempio iniziale il limite:

$$\lim_{x\to+\infty}\frac{1}{x^2-3x}$$

Possiamo calcolarlo con Maxima come segue:

```
(C1) f(x):=1/(x^2-3*x);
```

(D1) $f(x):=\frac{1}{x^2-3\,x}$

```
(C2) limit(f(x),x,inf);
```

(D2) 0

dove osserviamo che `inf` indica $+\infty$. Un altro esempio è costituito dal limite:

$$\lim_{x\to-\infty}\frac{e^x-e^{-x}}{e^x+e^{-x}}$$

che possiamo calcolare in Maxima come segue

```
(C7) limit((%e^x-%e^(-x))/(%e^x+%e^(-x)),x,minf);
```

(D7) -1

e dove osserviamo che con `minf` si indica $-\infty$. Oltre ai limiti all'infinito è possibile calcolare limiti il cui valore è infinito come ad esempio:

$$\lim_{x\to1}\frac{x}{(x-1)^2}$$

```
(C11) limit(x/(x-1)^2,x,1);
```

(D11) ∞

dove notiamo che con ∞ Maxima denota $+\infty$. Maxima non utilizza il concetto di infinito senza segno e quindi limiti come:

$$\lim_{x\to1}\frac{x}{x-1}$$

```
(C13) limit(x/(x-1),x,1);
```

(D13) UND

risultano indeterminati (UND). Possiamo tuttavia calcolare il valore dei limiti:

$$\lim_{x\to1^+}\frac{x}{x-1} \qquad \lim_{x\to1^-}\frac{x}{x-1}$$

ottenendo rispettivamente i valori

```
(C14) limit(x/(x-1),x,1,plus);
```

(D14) ∞

```
(C15) limit(x/(x-1),x,1,minus);
```

(D15) $-\infty$

3. Derivate

Attraverso il comando `diff` possiamo calcolare con Maxima le derivate di una funzione reale di variabile reale. Consideriamo ad esempio la funzione:

$$f(x)=e^{\sqrt{x}+1}$$

La sua derivata prima si ottiene scrivendo

```
(C16) diff(%e^(sqrt(x)-1),x);
```

(D16) $\frac{e^{\sqrt{x}-1}}{2\sqrt{x}}$

mentre la derivata seconda si ottiene scrivendo

```
(C17) diff(%e^(sqrt(x)-1),x,2);
```

(D17) $\frac{e^{\sqrt{x}-1}}{4\,x} - \frac{e^{\sqrt{x}-1}}{4\,x^{\frac{3}{2}}}$

Possiamo valutare anche il valore della derivata in un punto. Se ad esempio, data la funzione $f(x)=\sin(x^2+3x)$, si volesse calcolare $f'(0)$ si può procedere come segue:

```
(C1) f(x):=sin(x^2+3*x);
```

(D1) $f(x) := \sin\left(x^2+3\,x\right)$

```
(C2) derivata1: diff(f(x),x);
```

(D2) $(2\,x+3)\cos\left(x^2+3\,x\right)$

```
(C3) ev(derivata1,x=0);
```

(D3) 3

dove il comando `ev` (evaluate) valuta il valore di `derivata1` in $x=0$.

4. Applicazioni del calcolo differenziale

Vediamo ora come Maxima permetta di risolvere alcuni problemi in cui vengono applicati i concetti fondamentali di limite e derivata.

4.1. Analisi di costi

Consideriamo la funzione

$$m(x)=\frac{x}{100}+\frac{1600}{x}+\frac{1}{50} \qquad x>0$$

che esprime il costo medio *di una merce in funzione della quantità prodotta x. I problemi da risolvere sono due:*

1. *per quale valore di x il costo medio è minimo?*
2. *in corrispondenza di questo valore quanto vale il costo medio della merce?*

Iniziamo inserendo in Maxima la funzione costo medio:

```
(C1) m(x):=x/100+1600/x+1/50;
```

(D1) $m(x) := \frac{x}{100} + \frac{1600}{x} + \frac{1}{50}$

Possiamo quindi calcolare la derivata prima di $m(x)$ utilizzando il comando `diff`:

```
(C2) diff(m(x),x);
```

(D2) $\frac{1}{100} - \frac{1600}{x^2}$

Attraverso il comando `ratsimp` possiamo riscrivere la derivata prima in forma più espressiva:

```
(C3) ratsimp(%);
```

(D3) $\frac{x^2 - 160000}{100\, x^2}$

I valori di x che annullano la derivata prima si ottengono risolvendo la semplice equazione $x^2 - 160000 = 0$. In Maxima avremo:

```
(C4) num(%)=0;
```

(D4) $x^2 - 160000 = 0$

```
(C5) solve(%);
```

(D5) $[x = -400, x = 400]$

L'unico valore compatibile con i vincoli del problema è $x = 400$ che rappresenta la soluzione al primo quesito posto. La risposta al secondo quesito è ora immediata:

```
(C6) m(400);
```

(D6) $\frac{401}{50}$

4.2. Conti bancari

L'andamento del conto bancario del signor X, in migliaia di Euro, è descritto negli ultimi sette anni dalla funzione:

$$c(t) = t^3 - 12\, t^2 + 35\, t \qquad\qquad t \in [0, 7]$$

a) *In quali anni il conto bancario del signor X assume il valore 0?*

b) *In quale periodo il conto bancario di mister X è in rosso ($c(t) < 0$)?*

c) *In quale anno il conto bancario assume il valore massimo e a quanto ammonta* ***circa*** *il suo valore?*

d) *Studiare il segno della derivata seconda di $c(t)$.*

e) *Tracciare il grafico di $c(t)$ nel piano cartesiano (O, t, y).*

Possiamo risolvere questo problema con Maxima anche se in esso non possiamo risolvere le disequazioni.

```
(C1) c(t):=t^3-12*t^2+35*t;
```

(D1) $C(t) := t^3 - 12\, t^2 + 35\, t$

(C2) `solve(c(t)=0);`

(D2) $[t = 7, t = 5, t = 0]$

(C3) `derivata: diff(c(t),t);`

(D3) $3\, t^2 - 24\, t + 35$

(C4) `solve(derivata=0);`

(D4) $$\left[t = -\frac{\sqrt{39} - 12}{3}, t = \frac{\sqrt{39} + 12}{3}\right]$$

(C5) `%,numer;`

(D5) $[t = 1.918334000533867, t = 6.081665999466132]$

(C6) `diff(c(t),t,2);`

(D6) $6\, t - 24$

(C7) `solve(%);`

(D7) $[t = 4]$

(C8) `c(4);`

(D8) 12

Maxima ci permette innanzitutto di rispondere alla prima domanda in quanto gli anni in cui il conto è nullo sono 0, 5 e 7. Dalla forma della derivata prima si comprende che il punto di massimo relativo è $t = (12 - \sqrt{39})/3$ e che il conto in quell'anno vale circa 1918.33 Euro. Il conto risulta in rosso nel periodo che va da 5 a 7 anni. Il grafico della funzione ha la concavità rivolta verso il basso per $t \in [0, 4]$ e verso l'alto per $t \in [4, 7]$.

(C9) `plot2d(c(t),[t,0,7]);`



Figura 1. Andamento del conto bancario

4.3. Popolazione di topi

Una popolazione di topi cresce con una legge del tipo:

$$n(t) = \frac{5000}{1 + 2\,e^{-t}}$$

dove $n(t)$ indica il numero medio di topi e $t \geq 0$ indica il tempo. Si chiede:

1. *tracciare il grafico della funzione $n(t)$;*
2. *determinare il numero medio nel lungo periodo;*
3. *calcolare per quale valore di t si ha il massimo tasso di crescita $n'(t)$.*

Soluzione. La risposta alla prima domanda è lasciata al lettore. Per rispondere alla seconda domanda è sufficiente calcolare il limite:

$$\lim_{t \to +\infty} \frac{5000}{1 + 2\,e^{-t}}$$

```
(C1) n(t):=5000/(1+2*%e^(-t));
```

(D1) $n(t) := \frac{5000}{1 + 2\,e^{-t}}$

```
(C2) limit(n(t),t,inf);
```

(D4) 5000

Per rispondere alla terza domanda è necessario calcolare gli zeri della derivata seconda $n''(t)$:

```
(C5) diff(n(t),t,2);
```

(D5) $\frac{40000\,e^{-2t}}{(2\,e^{-t} + 1)^3} - \frac{10000\,e^{-t}}{(2\,e^{-t} + 1)^2}$

```
(C6) ratsimp(%);
```

(D6) $-\frac{10000\,e^{2t} - 20000\,e^{t}}{e^{3t} + 6\,e^{2t} + 12\,e^{t} + 8}$

```
(C7) num(%)=0;
```

(D7) $20000\,e^{t} - 10000\,e^{2t} = 0$

```
(C8) %/(10000*%e^t);
```

(D8) $\frac{e^{-t}\left(20000\,e^{t} - 10000\,e^{2t}\right)}{10000} = 0$

```
(C9) expand(%);
```

(D9) $2 - e^{t} = 0$

```
(C10) solve(%);
```

(D10) $[t = \log 2]$

L'unico zero della derivata seconda è $\log 2$ che in effetti (verificarlo!) è un punto di massimo della funzione tasso di crescita:

(C11) `diff(n(t),t);`

(D11) $\dfrac{10000\, e^{-t}}{(2\, e^{-t} + 1)^2}$

5. Integrali

Attraverso il comando `integrate` possiamo calcolare in Maxima integrali definiti o indefiniti come ad esempio:

$$\int \frac{1 - 3\sin^2 x}{3\sin^3 x}\, dx$$

(C19) `g(x):=(1-3*sin(x)^2)/(3*sin(x)^3);`

(D19) $g(x) := \dfrac{1 - 3\sin^2 x}{3\sin^3 x}$

(C20) `integrate(g(x),x);`

(D20) $\dfrac{\frac{5\log(\cos x + 1)}{4} - \frac{5\log(\cos x - 1)}{4} + \frac{\cos x}{2\cos^2 x - 2}}{3}$

oppure come

$$\int_0^2 \frac{x^2 - 2}{x^2 + 3x + 5}\, dx$$

(C21) `integrate((x^2-2)/(x^2+3*x+5),x,0,2);`

(D21) $\dfrac{10\sqrt{11}\arctan\left(\frac{3\sqrt{11}}{11}\right) + 33\log 5}{22} - \dfrac{10\sqrt{11}\arctan\left(\frac{7\sqrt{11}}{11}\right) + 33\log 15 - 44}{22}$

Data la complicazione del risultato può essere utile stimare il valore del risultato

(C22) `%,numer;`

(D22) -0.24040359145752